Techniques of Water-Resources Investigations of the United States Geological Survey

Chapter B7

ANALYTICAL SOLUTIONS FOR ONE-, TWO-, AND THREE-DIMENSIONAL SOLUTE TRANSPORT IN GROUND-WATER SYSTEMS WITH UNIFORM FLOW

By Eliezer J. Wexler

Book 3

APPLICATIONS OF HYDRAULICS

Attachment 2.—Program source-code listings

FINITE
SEMINF
POINT2
STRIPF
STRIPI
GAUSS
POINT3
PATCHF
PATCHI

```
C
C       ******************************************************
C       *                                                    *
C       *                  **** FINITE ****                  *
C       *                                                    *
C       * ONE-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL *
C       *                                                    *
C       *   FOR A FINITE SYSTEM WITH A FIRST- OR THIRD-TYPE  *
C       *                                                    *
C       *              BOUNDARY CONDITION AT X=0             *
C       *                                                    *
C       *            VERSION CURRENT AS OF 04/01/90          *
C       *                                                    *
C       ******************************************************
C
C     THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C     GREATER THAN THOSE GIVEN HERE.
C       MAXX = MAXIMUM NUMBER OF X-VALUES
C       MAXT = MAXIMUM NUMBER OF TIME VALUES
C       MAXRT = MAXIMUM NUMBER OF ROOTS USED IN THE SERIES SUMMATION
      PARAMETER MAXX=100,MAXT=20,MAXRT=1000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      REAL XP(MAXX),CP(MAXX),TP,XSCLP
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      CHARACTER*1 IERR(MAXX,MAXT)
      DIMENSION CXT(MAXX,MAXT),X(MAXX),T(MAXT)
      DIMENSION ROOT(MAXRT)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO       SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX       LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     VX       GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK       FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X        X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T        TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN       NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXT      SOLUTE CONCENTRATION C(X,T) [M/L**3]
C     XL       LENGTH OF THE FLOW SYSTEM [L]
C     ROOT(N)  ROOTS OF EQ. USED IN INFINITE SERIES SUMMATION
C
C     NBC      SOURCE BOUNDARY CONDITION TYPE (1 OR 3)
C     NX       NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT       NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     NROOT    NUMBER OF ROOTS USED IN INFINITE SERIES SUMMATION
C     IPLT     PLOT CONTROL. IF IPLT>0, CONCENTRATION PROFILES ARE PLOTT
C
```

```
C        CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C      CUNITS   UNITS OF CONCENTRATION (M/L**3)
C      VUNITS   UNITS OF GROUND-WATER VELOCITY (L/T)
C      DUNITS   UNITS OF DISPERSION COEFFICIENT (L**2/T)
C      KUNITS   UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C      LUNITS   UNITS OF LENGTH (L)
C      TUNITS   UNITS OF TIME (T)
C
C        DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C        READ IN MODEL PARAMETERS
      READ(IN,101) NBC,NX,NT,NROOT,IPLT
      IF(NBC.EQ.1) WRITE(IO,202)
      IF(NBC.EQ.3) WRITE(IO,203)
      WRITE(IO,205) NX,NT,NROOT
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) C0,VX,DX,DK,XL,XSCLP
      WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DK,KUNITS,XL,LUNITS,
     1 XSCLP
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
C
C        GET EIGENVALUES (BETA) USED IN SERIES SUMMATION BY SOLVING FOR
C        THE POSITIVE ROOTS OF: BETA*COTAN(BETA)+VX*XL/(2*DX)=0.0
C        FOR A FIRST-TYPE SOURCE BOUNDARY CONDITION,
C        OR: BETA*COTAN(BETA)-BETA**2*DX/(VX*XL)+VX*XL/(4*DX)=0.0
C        FOR A THIRD-TYPE SOURCE BOUNDARY CONDITION.
C
      IF (NBC.EQ.1) THEN
        C=VX*XL/(2.0D0*DX)
        CALL ROOT1(C,ROOT,NROOT)
      ELSE
        A=0.250D0*VX*XL/DX
        C=DX/(XL*VX)
        CALL ROOT3(A,C,ROOT,NROOT)
      END IF
C
C        BEGIN TIME LOOP
      DO 40 IT=1,NT
C
C        BEGIN X-COORDINATE LOOP
      DO 50 IX=1,NX
C
C        CALL ROUTINE TO CALCULATE NORMALIZED CONCENTRATION
C        BASED ON TYPE OF BOUNDARY CONDITION SPECIFIED
      IF(NBC.EQ.1) CALL CNRML1(XL,T(IT),X(IX),DX,VX,DK,ROOT,CN,NROOT,
```

```
     1 IERR(IX,IT))
      IF(NBC.EQ.3) CALL CNRML3(XL,T(IT),X(IX),DX,VX,DK,ROOT,CN,NROOT,
     1 IERR(IX,IT))
      CXT(IX,IT)=CN*C0
 50   CONTINUE
C
C       CONVERT X AND C TO SINGLE PRECISION AND DIVIDE BY C0 TO
C       PLOT NORMALIZED CONCENTRATION PROFILE FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 40
      DO 60 I=1,NX
      XP(I)=SNGL(X(I))
 60   CP(I)=SNGL(CXT(I,IT)/C0)
      TP=SNGL(T(IT))
      CALL PLOT1D(XP,CP,NX,TP,IT,NT,TUNITS,LUNITS,XSCLP)
 40   CONTINUE
C
C       PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NT-1)/9
      DO 80 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) TUNITS
      IF(NP.NE.1) WRITE(IO,231) TUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NT) NP2=NT-NP1
      WRITE(IO,235) (T(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXT(IX,NP1+J),IERR(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) TUNITS
      WRITE(IO,235) (T(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
 70   IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
 80   CONTINUE
C
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C        FORMAT STATEMENTS
 101  FORMAT(20I4)
 105  FORMAT(8A10)
 110  FORMAT(8F10.0)
 201  FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE ONE-DIMENSIONAL'/
     1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE-TRANSPORT EQUATION'/
     2 1H ,36X,'FOR A SYSTEM OF FINITE LENGTH'///1H0,40X,'INPUT DATA'/
     3 1H ,40X,10(1H-))
 202  FORMAT(1H0,25X,'FIRST-TYPE BOUNDARY CONDITION AT X = 0.0')
 203  FORMAT(1H0,25X,'THIRD-TYPE BOUNDARY CONDITION AT X = 0.0')
 205  FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,'NUMBER OF ROOTS ',
     2 'USED IN INFINITE SERIES SUMMATION (NROOT) = ',I4)
 210  FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (C0) =',
```

```
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'FIRST-ORDER SOLUTE-DECAY RATE (DK) =',1P1E13.6,1X,A10/
     5 1H ,25X,'LENGTH OF FINITE FLOW SYSTEM (XL) =',1P1E13.6,1X,A10/
     6 1H ,25X,'PLOT SCALING FACTOR (XSCLP) =',1P1E13.6)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AS A FUNCTION OF TIME',
     1 15X,'*  INDICATES SOLUTION DID NOT CONVERGE'/
     2 1H0,25X,'TIME VALUES, IN ',A10)
231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AS A FUNCTION OF TIME =',
     1 5X,'(CONTINUED)'/
     2 1H0,25X,'TIME VALUES, IN ',A10)
235   FORMAT(1H ,20X,9F12.4)
236   FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240   FORMAT(1H ,5X,F12.4,2X,'! ',9(F11.5,A1))
241   FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRML1(XL,T,X,D,V,DK,ROOT,CN,NROOT,IERR)
      IMPLICIT DOUBLE PRECISION  (A-H,O-Z)
      CHARACTER*1 IERR
      DIMENSION ROOT(NROOT)
C
C        SOLUTION FOR THE ONE-DIMENSIONAL SOLUTE-TRANSPORT EQUATION
C        FOR A SYSTEM OF FINITE LENGTH WITH A FIRST-TYPE SOURCE
C        BOUNDARY CONDITION. VALUE RETURNED IS THE NORMALIZED SOLUTE
C        CONCENTRATION AT A GIVEN X-COORDINATE AND TIME VALUE.
C        FOR NO SOLUTE DECAY, A SIMPLIFIED SOLUTION IS USED.
C
      IERR=' '
      XL2=XL*XL
      V2D=V/(2.0D0*D)
      VX2D=V2D*X
      VL2D=V2D*XL
      VL2D2=VL2D*VL2D
      DKL2D=DK*XL*XL/D
      VSQT4D=V*V*T/(4.0D0*D)
      IF(DK.EQ.0.0D0) GO TO 20
C
C        BEGIN SERIES SUMMATION FOR SOLUTE WITH DECAY
      SIGMA=0.0
      DO 10 N=1,NROOT
      BETA=ROOT(N)
      BETA2=BETA*BETA
C
C        TERM 1
      X1=(BETA2+VL2D2)*DEXP(-BETA2*D*T/XL2)
```

```
C
C       TERM 2
      DENOM=(BETA2+VL2D2+VL2D)*(BETA2+VL2D2+DKL2D)
      X2=BETA*DSIN(BETA*X/XL)/DENOM
      SIGMA=SIGMA+X1*X2
C
C     CHECK FOR CONVERGENCE OF SERIES
      IF(N.GT.25 .AND. DABS(X1*X2).LT.1.0D-14) GO TO 15
10    CONTINUE
      IERR='*'
15    CONTINUE
C
C       TERM 3
      U=DSQRT(V*V+4.0D0*DK*D)
      VMU=V-U
      VPU=V+U
      VUPM=(U-V)/VPU
      D2=D*2.0D0
      X3=DEXP(VMU*X/D2)+VUPM*DEXP((VPU*X-2.0D0*U*XL)/D2)
      X3=X3/(1.0D0+VUPM*DEXP(-U*XL/D))
      CN=X3-2.0D0*DEXP(VX2D-VSQT4D-DK*T)*SIGMA
      RETURN
C
C       BEGIN SERIES SUMMATION FOR SOLUTE WITH NO DECAY
20    SIGMA=0.0
      DO 30 N=1,NROOT
      BETA=ROOT(N)
      BETA2=BETA*BETA
C
C       TERM 1
      DENOM=BETA2+VL2D2+VL2D
      X1=BETA*DSIN(BETA*X/XL)*DEXP(-BETA2*D*T/XL2)
      X1=X1/DENOM
      TERM=X1
      SIGMA=SIGMA+X1
      IF(N.GT.25 .AND. DABS(X1).LT.1.0D-14) GO TO 35
30    CONTINUE
      IERR='*'
35    CONTINUE
      CN=1.0D0-2.0D0*DEXP(VX2D-VSQT4D)*SIGMA
      RETURN
      END
      SUBROUTINE CNRML3(XL,T,X,D,V,DK,ROOT,CN,NROOT,IERR)
      IMPLICIT DOUBLE PRECISION  (A-H,O-Z)
      CHARACTER*1 IERR
      DIMENSION ROOT(NROOT)
C
C       SOLUTION FOR THE ONE DIMENSIONAL SOLUTE-TRANSPORT EQUATION
C       FOR A SYSTEM OF FINITE LENGTH WITH A THIRD-TYPE SOURCE
C       BOUNDARY CONDITION. VALUE RETURNED IS THE NORMALIZED SOLUTE
C       CONCENTRATION AT A GIVEN X-COORDINATE AND TIME VALUE.
C       FOR NO SOLUTE DECAY, A SIMPLIFIED SOLUTION IS USED.
C
```

```
      IERR=' '
      XL2=XL*XL
      V2D=V/(2.0D0*D)
      VLD=V*XL/D
      VX2D=V2D*X
      VL2D=V2D*XL
      VL2D2=VL2D*VL2D
      DKL2D=DK*XL*XL/D
      VSQT4D=V*V*T/(4.0D0*D)
      IF(DK.EQ.0.0D0) GO TO 20
C
C       BEGIN SERIES SUMMATION FOR SOLUTE WITH DECAY
      SIGMA=0.0
      DO 10 N=1,NROOT
      BETA=ROOT(N)
      BETA2=BETA*BETA
C
C       TERM 1
      BETAXL=BETA*X/XL
      X1=BETA*(BETA*DCOS(BETAXL)+VL2D*DSIN(BETAXL))
C
C       TERM 2
      DENOM=(BETA2+VL2D2+VLD)*(BETA2+VL2D2+DKL2D)
      X2=DEXP(-BETA2*D*T/XL2)/DENOM
      SIGMA=SIGMA+X1*X2
C
C     CHECK FOR CONVERGENCE OF SERIES
      IF(N.GT.25 .AND. DABS(X1*X2).LT.1.0D-14) GO TO 15
10    CONTINUE
      IERR='*'
15    CONTINUE
C
C       TERM 3
      U=DSQRT(V*V+4.0D0*DK*D)
      VMU=V-U
      VPU=V+U
      VUPM=(U-V)/VPU
      D2=D*2.0D0
      X3=DEXP(VMU*X/D2)+VUPM*DEXP((VPU*X-2.0D0*U*XL)/D2)
      X3=2.0D0*V*X3/(VPU+VMU*VUPM*DEXP(-U*XL/D))
      CN=X3-2.0D0*VLD*DEXP(VX2D-VSQT4D-DK*T)*SIGMA
      RETURN
C
C       BEGIN SERIES SUMMATION FOR SOLUTE WITH NO DECAY
20    SIGMA=0.0
      DO 30 N=1,NROOT
      BETA=ROOT(N)
      BETA2=BETA*BETA
C
C       TERM 1
      BETAXL=BETA*X/XL
      X1=BETA*(BETA*DCOS(BETAXL)+VL2D*DSIN(BETAXL))
C
```

```
C       TERM 2
      DENOM=(BETA2+VL2D2+VLD)*(BETA2+VL2D2)
      X2=DEXP(-BETA2*D*T/XL2)/DENOM
C
      SIGMA=SIGMA+X1*X2
      IF(N.GT.25 .AND. DABS(X1*X2).LT.1.0D-14) GO TO 35
30    CONTINUE
      IERR='*'
35    CONTINUE
C
      CN=1.0D0-2.0D0*VLD*DEXP(VX2D-VSQT4D)*SIGMA
      RETURN
      END
      SUBROUTINE ROOT1 (C,ROOT,NROOT)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION ROOT(NROOT)
      COMMON /IOUNIT/ IN,IO
      DATA MAXIT,EPS/50,1.0D-10/
C
C       THIS ROUTINE CALCULATES ROOTS OF THE EQUATION: B*COTAN(B)+C=0
C       USING NEWTON'S SECOND-ORDER METHOD.
C
C                      PROGRAM VARIABLES
C       MAXIT     MAXIMUM NUMBER OF ITERATIONS ALLOWED IN ROOT SEARCH
C       EPS       CONVERGENCE CRITERION
C       F1,F2     1ST AND 2ND DERIVATIVES OF THE EQUATION
C       H         SECOND-ORDER CORRECTION FACTOR
C
C       FIRST ROOT LIES BETWEEN PI/2 AND PI. START WITH .75*PI
      PI=3.14159265359D0
      ROOT(1)=0.750D0*PI
C
C       START LOOP FOR EACH ROOT SEARCH
      DO 10 N=1,NROOT
C
C       BEGIN ITERATIVE LOOP
      DO 20 I=1,MAXIT
      X=ROOT(N)
      SINX2=DSIN(X)*DSIN(X)
      COTX=1.0D0/DTAN(X)
      F=X*COTX+C
C       IF F IS 0.0, EXACT ROOT HAS BEEN FOUND
      IF(F.EQ.0.0) GO TO 30
      F1=COTX-X/SINX2
      F2=-1.0D0/SINX2-(SINX2-X*DSIN(X*2.0D0))/(SINX2*SINX2)
      H=(F2/2.0D0)/F1-F1/F
      H=1.0D0/H
      ROOT(N)=X+H
C
C     CHECK FOR CONVERGENCE. IF NOT ACHIEVED, RE-ITERATE
      IF(DABS(H).LT.EPS) GO TO 30
   20 CONTINUE
      WRITE(IO,201) MAXIT,N
```

```
      STOP
C
C       NEXT ROOT IS ABOUT PI GREATER THAN LAST ROOT
   30 IF(N.NE.NROOT) ROOT(N+1)=ROOT(N)+PI
   10 CONTINUE
      RETURN
C
C       FORMAT STATEMENTS
201   FORMAT(1H ,5X,'**** WARNING ****  ROOT SEARCH ROUTINE DID NOT',
     1 'CONVERGE AFTER ',I4,'ITERATIONS WHILE SEARCHING FOR ROOT',I5)
      END
      SUBROUTINE ROOT3 (A,C,ROOT,NROOT)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION ROOT(NROOT)
      COMMON /IOUNIT/ IN,IO
      DATA MAXIT,EPS/50,1.0D-10/
C
C       THIS ROUTINE CALCULATES ROOTS OF THE EQ: B*COTAN(B)-C*B**2+A=0
C       USING NEWTON'S SECOND-ORDER METHOD.
C
C                      PROGRAM VARIABLES
C       MAXIT    MAXIMUM NUMBER OF ITERATIONS ALLOWED IN ROOT SEARCH
C       EPS      CONVERGENCE CRITERION
C       F1,F2    1ST AND 2ND DERIVATIVES OF THE EQUATION
C       H        SECOND-ORDER CORRECTION FACTOR
C
C       FIRST ROOT LIES BETWEEN 0.0 AND PI. START WITH 0.5*PI
      PI=3.14159265359D0
      ROOT(1)=0.50D0*PI
C
C       START LOOP FOR EACH ROOT SEARCH
      DO 10 N=1,NROOT
C
C       BEGIN ITERATIVE LOOP
      DO 20 I=1,MAXIT
      X=ROOT(N)
      SINX2=DSIN(X)*DSIN(X)
      COTX=1.0D0/DTAN(X)
      F=X*COTX-C*X*X+A
C       IF F IS 0.0, EXACT ROOT HAS BEEN FOUND
      IF(F.EQ.0.0) GO TO 30
      F1=COTX-X/SINX2-(2.0D0*C*X)
      F2=-1.0D0/SINX2-(SINX2-X*DSIN(X*2.0D0))/(SINX2*SINX2)-2.0D0*C
      H=(F2/2.0D0)/F1-F1/F
      H=1.0D0/H
      ROOT(N)=X+H
C
C     CHECK FOR CONVERGENCE. IF NOT ACHIEVED, RE-ITERATE
      IF(DABS(H).LT.EPS) GO TO 30
   20 CONTINUE
      WRITE(IO,201) MAXIT,N
      STOP
C
```

```
C        NEXT ROOT IS ABOUT PI GREATER THAN LAST ROOT
   30 IF(N.NE.NROOT) ROOT(N+1)=ROOT(N)+PI
   10 CONTINUE
      RETURN
C
C        FORMAT STATEMENTS
201   FORMAT(1H ,5X,'**** WARNING ****  ROOT SEARCH ROUTINE DID NOT',
     1 'CONVERGE AFTER ',I4,'ITERATIONS WHILE SEARCHING FOR ROOT',I5)
      END
```

```
C
C      *******************************************************
C      *                                                     *
C      *                  **** SEMINF ****                   *
C      *                                                     *
C      * ONE-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL *
C      *                                                     *
C      *   FOR A SEMI-INFINITE SYSTEM WITH A FIRST-TYPE OR   *
C      *                                                     *
C      *         THIRD-TYPE BOUNDARY CONDITION AT X=0        *
C      *                                                     *
C      *            VERSION CURRENT AS OF 04/01/90           *
C      *                                                     *
C      *******************************************************
C
C       THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C       GREATER THAN THOSE GIVEN HERE.
C         MAXX = MAXIMUM NUMBER OF X-VALUES
C         MAXT = MAXIMUM NUMBER OF TIME VALUES
      PARAMETER MAXX=100,MAXT=20
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      REAL XP(MAXX),CP(MAXX),TP,XSCLP
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      DIMENSION CXT(MAXX,MAXT),X(MAXX),T(MAXT)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO       SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX       LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     VX       GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK       FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X        X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T        TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN       NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXT      SOLUTE CONCENTRATION C(X,T) [M/L**3]
C
C     NBC      SOURCE BOUNDARY CONDITION TYPE (1 OR 3)
C     NX       NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT       NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     IPLT     PLOT CONTROL. IF IPLT>0, CONCENTRATION PROFILES ARE PLOTT
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS   UNITS OF CONCENTRATION (M/L**3)
C     VUNITS   UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS   UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS   UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS   UNITS OF LENGTH (L)
```

```
C     TUNITS   UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NBC,NX,NT,IPLT
      IF(NBC.EQ.1) WRITE(IO,202)
      IF(NBC.EQ.3) WRITE(IO,203)
      WRITE(IO,205) NX,NT
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) C0,VX,DX,DK,XSCLP
      WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DK,KUNITS,XSCLP
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
C
C       BEGIN TIME LOOP
      DO 40 IT=1,NT
C
C       BEGIN X-COORDINATE LOOP
      DO 50 IX=1,NX
C
C       CALL ROUTINE TO CALCULATE NORMALIZED CONCENTRATION
C       BASED ON TYPE OF BOUNDARY CONDITION SPECIFIED
      IF(NBC.EQ.1) CALL CNRML1(DK,T(IT),X(IX),DX,VX,CN)
      IF(NBC.EQ.3) CALL CNRML3(DK,T(IT),X(IX),DX,VX,CN)
      CXT(IX,IT)=CN*C0
   50 CONTINUE
C
C       CONVERT X AND C TO SINGLE PRECISION AND DIVIDE BY C0 TO
C       PLOT NORMALIZED CONCENTRATION PROFILE FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 40
      DO 60 I=1,NX
      XP(I)=SNGL(X(I))
   60 CP(I)=SNGL(CXT(I,IT)/C0)
      TP=SNGL(T(IT))
      CALL PLOT1D(XP,CP,NX,TP,IT,NT,TUNITS,LUNITS,XSCLP)
   40 CONTINUE
C
C       PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NT-1)/9
      DO 80 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) TUNITS
      IF(NP.NE.1) WRITE(IO,231) TUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NT) NP2=NT-NP1
```

```
      WRITE(IO,235) (T(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXT(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) TUNITS
      WRITE(IO,235) (T(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
70    IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
80    CONTINUE
C
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C        FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
110   FORMAT(8F10.0)
201   FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE ONE-DIMENSIONAL'/
     1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,38X,'FOR A SEMI-INFINITE SYSTEM'///1H0,40X,'INPUT DATA'/
     3 1H ,40X,10(1H-))
202   FORMAT(1H0,25X,'FIRST-TYPE BOUNDARY CONDITION AT X = 0.0')
203   FORMAT(1H0,25X,'THIRD-TYPE BOUNDARY CONDITION AT X = 0.0')
205   FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF TIME VALUES (NT) = ',I4)
210   FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (C0) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10/
     5 1H ,25X,'PLOT SCALING FACTOR (XSCLP) =',1P1E13.6)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AS A FUNCTION OF TIME'/
     2 1H0,25X,'TIME VALUES, IN ',A10)
231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AS A FUNCTION OF TIME =',
     1 5X,'(CONTINUED)'/
     2 1H0,25X,'TIME VALUES, IN ',A10)
235   FORMAT(1H ,20X,9F12.4)
236   FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240   FORMAT(1H ,5X,F12.4,2X,'! ',9F12.5)
241   FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRML1(DK,T,X,D,V,CN)
      IMPLICIT DOUBLE PRECISION  (A-H,O-Z)
C
```

```
C       SOLUTION FOR THE ONE-DIMENSIONAL SOLUTE TRANSPORT EQUATION
C       FOR A SEMI-INFINITE SYSTEM WITH A FIRST-TYPE SOURCE
C       BOUNDARY CONDITION. VALUE RETURNED IS THE NORMALIZED SOLUTE
C       CONCENTRATION AT A GIVEN X-COORDINATE AND TIME.
C       FOR NO SOLUTE DECAY, A SIMPLIFIED SOLUTION IS USED.
C
      ALPHA=2.0D0*DSQRT(D*T)
      U=DSQRT(V*V+4.0D0*D*DK)
      X2D=X/(2.0D0*D)
C
C       SOLUTION WITH SOLUTE DECAY
      IF(DK.EQ.0.0) GO TO 10
C
C       TERM 1
      X1=X2D*(V-U)
      Y1=(X-U*T)/ALPHA
      CALL EXERFC(X1,Y1,Z1)
C
C       TERM 2
      X2=X2D*(V+U)
      Y2=(X+U*T)/ALPHA
      CALL EXERFC(X2,Y2,Z2)
      CN=(Z1+Z2)/(2.0D0)
      RETURN
C
C       SOLUTION WITH NO SOLUTE DECAY
C       TERM 1
10    Y1=(X-V*T)/ALPHA
      CALL EXERFC(0.0D0,Y1,Z1)
C
C       TERM 2
      X2=X*V/D
      Y2=(X+V*T)/ALPHA
      CALL EXERFC(X2,Y2,Z2)
      CN=(Z1+Z2)/2.0D0
      RETURN
      END
      SUBROUTINE CNRML3(DK,T,X,D,V,CN)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C       SOLUTION FOR THE ONE-DIMENSIONAL SOLUTE TRANSPORT EQUATION
C       FOR A SEMI-INFINITE SYSTEM WITH A THIRD-TYPE SOURCE
C       BOUNDARY CONDITION. VALUE RETURNED IS THE NORMALIZED SOLUTE
C       CONCENTRATION AT A GIVEN X-COORDINATE AND TIME.
C       FOR NO SOLUTE DECAY. A SIMPLIFIED SOLUTION IS USED.
C
      ALPHA=2.0D0*DSQRT(D*T)
      U=DSQRT(V*V+4.0D0*D*DK)
      X2D=X/(2.0D0*D)
      VXD=V*X/D
C
C       SOLUTION WITH SOLUTE DECAY
      IF(DK.EQ.0.0) GO TO 10
```

```
C
C       TERM 1
      X1=VXD-DK*T
      Y1=(X+V*T)/ALPHA
      CALL EXERFC(X1,Y1,Z1)
      Z1=Z1*2.0D0
C
C       TERM 2
      X2=X2D*(V-U)
      Y2=(X-U*T)/ALPHA
      CALL EXERFC(X2,Y2,Z2)
      Z2=Z2*(U/V-1.0D0)
C
C       TERM 3
      X3=X2D*(V+U)
      Y3=(X+U*T)/ALPHA
      CALL EXERFC(X3,Y3,Z3)
      Z3=Z3*(U/V+1.0D0)
      CN=V*V*(Z1+Z2-Z3)/(4.0D0*D*DK)
      RETURN
C
C       SOLUTION FOR NO SOLUTE DECAY
10    PI=3.14159265358979D0
C
C       TERM 1
      Y1=(X-V*T)/ALPHA
      CALL EXERFC(0.0D0,Y1,Z1)
      Z1=0.50D0*Z1
C
C       TERM 2
      X2=VXD
      Y2=(X+V*T)/ALPHA
      CALL EXERFC(X2,Y2,Z2)
      Z2=Z2*0.50D0*(1.0D0+V*(X+V*T)/D)
C
C       TERM 3
      Z3=DEXP(-1.0D0*Y1*Y1)
      Z3=Z3*V*DSQRT(T/(PI*D))
      CN=Z1-Z2+Z3
      RETURN
      END
```

```
C
C       ********************************************************
C       *                                                      *
C       *                  **** POINT2 ****                    *
C       *                                                      *
C       * TWO-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL  *
C       *                                                      *
C       *   FOR AN AQUIFER OF INFINITE AREAL EXTENT WITH A     *
C       *                                                      *
C       *  CONTINUOUS POINT SOURCE LOCATED AT X=XC AND Y=YC    *
C       *                                                      *
C       *       GROUND-WATER FLOW IN X-DIRECTION ONLY          *
C       *                                                      *
C       *           VERSION CURRENT AS OF 04/01/90             *
C       *                                                      *
C       ********************************************************
C
C     THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C     GREATER THAN THOSE GIVEN HERE.
C       MAXX = MAXIMUM NUMBER OF X-VALUES
C       MAXY = MAXIMUM NUMBER OF Y-VALUES
C       MAXT = MAXIMUM NUMBER OF TIME VALUES
C       MAXXY = MAXX * MAXY
C       MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      REAL XP,YP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C          NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C          MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C          LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO        SOLUTE CONCENTRATION IN INJECTED FLUID [M/L**3]
C     DX        LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY        TRANSVERSE DISPERSION COEFFICIENT [L**2/T]
C     VX        GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK        FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X         X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y         Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T         TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN        NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXY       SOLUTE CONCENTRATION C(X,Y,T) [M/L**3]
C     XC        X-COORDINATE OF POINT SOURCE [L]
C     YC        Y-COORDINATE OF POINT SOURCE [L]
C     QM        FLUID INJECTION RATE PER UNIT THICKNESS OF AQUIFER [L**2/
```

```
C                (UNITS MUST BE SAME AS DISPERSION COEFFICIENT)
C      POR      AQUIFER PORISITY [DIMENSIONLESS]
C
C      NX       NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C      NY       NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C      NT       NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C      NMAX     NUMBER OF TERMS USED IN GAUSS-LEGENDRE NUMERICAL
C               INTEGRATION TECHNIQUE (MUST EQUAL 4, 20, 60, 104 OR 256)
C
C      IPLT     PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C      XSCLP    SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C      YSCLP    SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C      DELTA    CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C        CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C      CUNITS   UNITS OF CONCENTRATION (M/L**3)
C      VUNITS   UNITS OF GROUND-WATER VELOCITY (L/T)
C      DUNITS   UNITS OF DISPERSION COEFFICIENT (L**2/T)
C      KUNITS   UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C      LUNITS   UNITS OF LENGTH (L)
C      TUNITS   UNITS OF TIME (T)
C
C        DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
       CALL OFILE
       CALL TITLE
       WRITE(IO,201)
C
C        READ IN MODEL PARAMETERS
       READ(IN,101) NX,NY,NT,NMAX,IPLT
       WRITE(IO,205) NX,NY,NT,NMAX
       READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
       READ(IN,110) C0,VX,DX,DY,DK
       WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DK,KUNITS
       READ(IN,110) XC,YC,QM,POR
       WRITE(IO,212) XC,LUNITS,YC,LUNITS,QM,DUNITS,POR
       READ(IN,110) (X(I),I=1,NX)
       WRITE(IO,215) LUNITS
       WRITE(IO,220) (X(I),I=1,NX)
       READ(IN,110) (Y(I),I=1,NY)
       WRITE(IO,216) LUNITS
       WRITE(IO,220) (Y(I),I=1,NY)
       READ(IN,110) (T(I),I=1,NT)
       WRITE(IO,225) TUNITS
       WRITE(IO,220) (T(I),I=1,NT)
       IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
       IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C        READ IN GAUSS-LEGENDRE POINTS AND WEIGHTING FACTORS
       CALL GLQPTS (NMAX)
C
C        BEGIN TIME LOOP
       DO 20 IT=1,NT
C
```

```
C         BEGIN X LOOP
        DO 40 IX=1,NX
        XX=X(IX)-XC
C
C       CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX)
        DO 50 IY=1,NY
        YY=Y(IY)-YC
        CALL CNRML2(QM,POR,DK,T(IT),XX,YY,DX,DY,VX,CN,NMAX)
        CXY(IX,IY)=C0*CN
 50     CONTINUE
 40     CONTINUE
C
C         PRINT OUT TABLES OF CONCENTRATION VALUES
        NPAGE=1+(NY-1)/9
        DO 60 NP=1,NPAGE
        IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,LUNITS
        IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,LUNITS
        NP1=(NP-1)*9
        NP2=9
        IF((NP1+NP2).GT.NY) NP2=NY-NP1
        WRITE(IO,235) (Y(NP1+J),J=1,NP2)
        WRITE(IO,236) CUNITS,LUNITS
        DO 70 IX=1,NX
        WRITE(IO,240) X(IX),(CXY(IX,NP1+J),J=1,NP2)
        IF(MOD(IX,45).NE.0) GO TO 70
        WRITE(IO,231) T(IT),TUNITS,LUNITS
        WRITE(IO,235) (Y(NP1+J),J=1,NP2)
        WRITE(IO,236) CUNITS,LUNITS
 70     IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
 60     CONTINUE
C
C         CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C         PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY C0 TO PLOT
C         CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
        IF(IPLT.LT.1) GO TO 20
        NXY=NX*NY
        DO 80 I=1,NX
        IP=(I-1)*NY
        XP(I)=SNGL(X(I))
        DO 80 J=1,NY
        IF(I.EQ.1) YP(J)=SNGL(Y(J))
        CP(IP+J)=SNGL(CXY(I,J)/C0)
 80     CONTINUE
        TP=SNGL(T(IT))
        NXY2=NXY*2
        CALL PLOT2D (XP,YP,CP,TP,DELTA,NX,NY,NXY,NXY2,IT,NT,IPLT,TUNITS,
      1 LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
 20     CONTINUE
        CLOSE (IN)
        CLOSE (IO)
        STOP
C
C         FORMAT STATEMENTS
```

```
101    FORMAT(20I4)
105    FORMAT(8A10)
110    FORMAT(8F10.0)
201    FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE TWO-DIMENSIONAL'/
      1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
      2 1H ,28X,'FOR AN AQUIFER OF INFINITE AREAL EXTENT WITH A'/
      3 1H ,31X,'CONTINUOUS POINT SOURCE AT X=0 AND Y=YC'
      4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205    FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
      1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
      2 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
      3 'NUMBER OF POINTS FOR NUMERICAL INTEGRATION (NMAX) = ',I4)
210    FORMAT(1H0,25X,'SOLUTE CONCENTRATION IN INJECTED FLUID (CO) =',
      1 1P1E13.6,1X,A10/1H ,25X,
      2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
      3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
      4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
      5 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212    FORMAT(1H0,25X,'AQUIFER IS OF INFINITE AREAL EXTENT'/1H ,25X,
      1 'CONTINUOUS POINT SOURCE IS LOCATED AT X =',1P1E13.6,1X,A10/1H ,
      1 63X,'Y =',1P1E13.6,1X,A10/1H ,25X,
      2 'FLUID INJECTION RATE PER UNIT THICHKNESS OF AQUIFER (QM) =',
      3 1P1E13.6,1X,A10/1H ,25X,'AQUIFER POROSITY (POR) =',1P1E13.6)
215    FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216    FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220    FORMAT(1H ,5X,8F12.4)
225    FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227    FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
      1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
      2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230    FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
      1 F12.4,1X,A10/
      2 1H0,25X,'Y-COORDINATE, IN ',A10)
231    FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
      1 F12.4,1X,A10,5X,'(CONTINUED)'/
      2 1H0,25X,'Y-COORDINATE, IN ',A10)
235    FORMAT(1H ,20X,9F12.4)
236    FORMAT(1H ,19X,'*',108(1H-)/
      1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
      2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240    FORMAT(1H ,5X,F12.4,2X,'!',9F12.5)
241    FORMAT(1H ,19X,'!')
       END
       SUBROUTINE CNRML2(QM,POR,DK,T,X,Y,DX,DY,VX,CN,NMAX)
       IMPLICIT DOUBLE PRECISION(A-H,O-Z)
       COMMON /IOUNIT/ IN,IO
       COMMON /GLPTS/ WN(256),ZN(256)
C
C        THIS ROUTINE CALCULATES SOLUTE CONCENTRATION AT X,Y BASED ON
C        THE ANALYTIC SOLUTION TO THE TWO-DIMENSIONAL ADVECTIVE-
```

```
C        DISPERSIVE SOLUTE TRANSPORT EQUATION FOR AN AQUIFER OF
C        INFINITE AREAL EXTENT WITH A CONTINUOUS POINT SOURCE LOCATED
C        AT X=XC AND Y=YC.  THE INTEGRAL FROM 0 TO T IS EVALUATED
C        USING A GAUSS-LEGENDRE QUADRATURE INTEGRATION TECHNIQUE.
C
      PI=3.14159265358979D0
      CN=0.0D0
C
C        FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C        START NUMERICAL INTEGRATION LOOP
      ALPHA=X*X/(4.0D0*DX)+Y*Y/(4.0D0*DY)
      BETA=VX*VX/(4.0D0*DX)+DK
      VX2D=VX*X/(2.0D0*DX)
      SUM=0.0D0
      DO 20 I=1,NMAX
C
C        SCALE THE GAUSS-LEGENDRE COEFFICIENTS TO ACCOUNT FOR THE
C        NON-NORMALIZED LIMITS OF INTEGRATION
      WI=WN(I)
      ZI=T*(ZN(I)+1.0D0)/2.0D0
C
C        TERM 1
      X1=-ALPHA/ZI-BETA*ZI
      X1=DEXP(X1)/ZI
      SUM=SUM+X1*WI
   20 CONTINUE
      SUM=SUM *T/2.0D0
      CN=QM*SUM*DEXP(VX2D)/(4.0D0*POR*PI*DSQRT(DX*DY))
      RETURN
      END
```

```
C
C        *******************************************************
C        *                                                     *
C        *                  **** STRIPF ****                   *
C        *                                                     *
C        * TWO-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL *
C        *                                                     *
C        *   FOR A SEMI-INFINITE AQUIFER WITH A FINITE WIDTH   *
C        *                                                     *
C        *     A STRIP SOURCE EXTENDS FROM Y1 TO Y2 AT X=0     *
C        *                                                     *
C        *        GROUND-WATER FLOW IN X-DIRECTION ONLY        *
C        *                                                     *
C        *            VERSION CURRENT AS OF 04/01/90           *
C        *                                                     *
C        *******************************************************
C
C       THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C       GREATER THAN THOSE GIVEN HERE.
C         MAXX = MAXIMUM NUMBER OF X-VALUES
C         MAXY = MAXIMUM NUMBER OF Y-VALUES
C         MAXT = MAXIMUM NUMBER OF TIME VALUES
C         MAXXY = MAXX * MAXY
C         MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      CHARACTER*1 IERR(MAXX,MAXY)
      REAL XP,YP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     C0        SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX        LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY        TRANSVERSE DISPERSION COEFFICIENT [L**2/T]
C     VX        GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK        FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X         X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y         Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T         TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN        NORMALIZED CONCENTRATION C/C0 [DIMENSIONLESS]
C     CXY       SOLUTE CONCENTRATION C(X,Y,T) [M/L**3]
C     W         AQUIFER WIDTH (AQUIFER EXTENDS FROM Y=0 TO Y=W) [L]
C     Y1        Y-COORDINATE OF LOWER LIMIT OF STRIP SOLUTE SOURCE [L]
```

```
C     Y2        Y-COORDINATE OF UPPER LIMIT OF STRIP SOLUTE SOURCE [L]
C
C     NX        NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NY        NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT        NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     NMAX      NUMBER OF TERMS USED IN INFINITE SERIES SUMMATION
C
C     IPLT      PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C     XSCLP     SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C     YSCLP     SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C     DELTA     CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS    UNITS OF CONCENTRATION (M/L**3)
C     VUNITS    UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS    UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS    UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS    UNITS OF LENGTH (L)
C     TUNITS    UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NX,NY,NT,NMAX,IPLT
      WRITE(IO,205) NX,NY,NT,NMAX
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) C0,VX,DX,DY,DK
      WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DK,KUNITS
      READ(IN,110) W,Y1,Y2
      WRITE(IO,212) W,LUNITS,Y1,LUNITS,Y2,LUNITS
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (Y(I),I=1,NY)
      WRITE(IO,216) LUNITS
      WRITE(IO,220) (Y(I),I=1,NY)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
      IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
      IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C       BEGIN TIME LOOP
      DO 20 IT=1,NT
C
C       BEGIN X LOOP
      DO 40 IX=1,NX
C
C     CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX)
      DO 50 IY=1,NY
```

```
      CALL CNRMLF(DK,T(IT),X(IX),Y(IY),W,Y1,Y2,DX,DY,
     1 VX,CN,NMAX,IERR(IX,IY))
      CXY(IX,IY)=C0*CN
50    CONTINUE
40    CONTINUE
C
C        PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),IERR(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
70    IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
60    CONTINUE
C
C        CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C        PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY C0 TO PLOT
C        CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 20
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
      CP(IP+J)=SNGL(CXY(I,J)/C0)
80    CONTINUE
      TP=SNGL(T(IT))
      NXY2=NXY*2
      CALL PLOT2D (XP,YP,CP,TP,DELTA,NX,NY,NXY,NXY2,IT,NT,IPLT,TUNITS,
     1 LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
20    CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C         FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
110   FORMAT(8F10.0)
201   FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE TWO-DIMENSIONAL'/
     1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,30X,'FOR A SEMI-INFINITE AQUIFER OF FINITE WIDTH'/
```

```
     3 1H ,26X,'WITH A FINITE-WIDTH (STRIP) SOLUTE SOURCE AT X=0.0'
     4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
 205  FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
     2 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
     3 'NUMBER OF TERMS IN INFINTE SERIES SUMMATION (NMAX) = ',I4)
 210  FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (CO) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
     5 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
 212  FORMAT(1H0,25X,'AQUIFER WIDTH (W) =',1P1E13.6,1X,A10/1H ,25X,
     1 'SOLUTE SOURCE IS LOCATED BETWEEN Y1 =',1P1E13.6,1X,A10/1H ,54X,
     2 'AND Y2 =',1P1E13.6,1X,A10)
 215  FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
 216  FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
 220  FORMAT(1H ,5X,8F12.4)
 225  FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
 227  FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
     1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
     2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
 230  FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10,15X,'*  INDICATES SOLUTION DID NOT CONVERGE'/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
 231  FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10,5X,'(CONTINUED)'/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
 235  FORMAT(1H ,20X,9F12.4)
 236  FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
 240  FORMAT(1H ,5X,F12.4,2X,'! ',9(F11.5,A1))
 241  FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRMLF(DK,T,X,Y,W,Y1,Y2,DX,DY,VX,CN,NMAX,IERR)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      CHARACTER*1 IERR
      COMMON /IOUNIT/ IN,IO
C
C       THIS ROUTINE CALCULATES THE NORMALIZED CONCENTRATION AT X,Y
C       BASED ON THE ANALYTIC SOLUTION TO THE TWO-DIMENSIONAL
C       ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION FOR A SEMI-
C       INFINITE AQUIFER WITH A FINITE WIDTH.  A FINITE-WIDTH (STRIP)
C       SOLUTE SOURCE EXTENDS FROM Y=Y1 TO Y=Y2. THE SOLUTION
C       CONTAINS AN INFINITE SERIES SUMMATION WHICH MAY TAKE A LARGE
C       NUMBER OF TERMS TO CONVERGE FOR SMALL VALUES OF X.
C
      PI=3.14159265358979D0
      CN=0.0D0
```

```
      IERR=' '
C
C       FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C       FOR X=0.0, CONCENTRATIONS ARE SPECIFIED BY BOUNDARY CONDITIONS
      IF(X.GT.0.0D0) GO TO 10
      IF(Y.GT.Y1 .AND. Y.LT.Y2) CN=1.0D0
      IF(Y.EQ.Y1) CN=0.50D0
      IF(Y.EQ.Y2) CN=0.50D0
      RETURN
C
C       BEGIN SUMMATION OF TERMS IN INFINITE SERIES
10    RTDXT=2.0D0*DSQRT(DX*T)
      SIGMA=0.0D0
      SUBTOT=0.0D0
      NMAX1=NMAX+1
      DO 20 NN=1,NMAX1
      N=NN-1
      ETA=N*PI/W
      PN=(Y2-Y1)/(2.0D0*W)
      IF(N.NE.0) PN=(DSIN(ETA*Y2)-DSIN(ETA*Y1))/(N*PI)
      COSRY=DCOS(ETA*Y)
      ALPHA=4.0D0*DX*(ETA*ETA*DY+DK)
      BETA=DSQRT(VX*VX+ALPHA)
      BETAT=BETA*T
C
C       CALCULATE TERM 1
      A1=X*(VX-BETA)/(2.0D0*DX)
      B1=(X-BETAT)/RTDXT
      CALL EXERFC(A1,B1,C1)
C
C       CALCULATE TERM 2
      A2=X*(VX+BETA)/(2.0D0*DX)
      B2=(X+BETAT)/RTDXT
      CALL EXERFC(A2,B2,C2)
C
C       ADD TERMS TO SUMMATION
      TERM=PN*COSRY*(C1+C2)
      SIGMA=SIGMA+TERM
C
C       CHECK FOR CONVERGENCE. BECAUSE SERIES OSCILLATES, CHECK
C       SUBTOTAL OF LAST 10 TERMS.
      SUBTOT=SUBTOT+TERM
      IF(MOD(NN,10).NE.0) GO TO 20
      IF(DABS(SUBTOT).LT.1.0D-12) GO TO 30
      SUBTOT=0.0D0
20    CONTINUE
      IERR='*'
30    CN=SIGMA
      RETURN
      END
```

```
C
C       ******************************************************
C       *                                                    *
C       *                  **** STRIPI ****                  *
C       *                                                    *
C       * TWO-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL *
C       *                                                    *
C       *    FOR A SEMI-INFINITE AQUIFER OF INFINITE WIDTH   *
C       *                                                    *
C       *     A STRIP SOURCE EXTENDS FROM Y1 TO Y2 AT X=0    *
C       *                                                    *
C       *        GROUND-WATER FLOW IN X-DIRECTION ONLY       *
C       *                                                    *
C       *            VERSION CURRENT AS OF 04/01/90          *
C       *                                                    *
C       ******************************************************
C
C     THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C     GREATER THAN THOSE GIVEN HERE.
C       MAXX = MAXIMUM NUMBER OF X-VALUES
C       MAXY = MAXIMUM NUMBER OF Y-VALUES
C       MAXT = MAXIMUM NUMBER OF TIME VALUES
C       MAXXY = MAXX * MAXY
C       MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      REAL XP,YP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     C0       SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX       LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY       TRANSVERSE DISPERSION COEFFICIENT [L**2/T]
C     VX       GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK       FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X        X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y        Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T        TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN       NORMALIZED CONCENTRATION C/C0 [DIMENSIONLESS]
C     CXY      SOLUTE CONCENTRATION C(X,Y,T) [M/L**3]
C     Y1       Y-COORDINATE OF LOWER LIMIT OF STRIP SOLUTE SOURCE [L]
C     Y2       Y-COORDINATE OF UPPER LIMIT OF STRIP SOLUTE SOURCE [L]
C     NX       NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
```

```
C      NY        NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C      NT        NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C      NMAX      NUMBER OF TERMS USED IN GAUSS-LEGENDRE NUMERICAL
C                INTEGRATION TECHNIQUE (MUST EQUAL 4, 20, 60, 104 OR 256)
C
C      IPLT      PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C      XSCLP     SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C      YSCLP     SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C      DELTA     CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C        CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C      CUNITS    UNITS OF CONCENTRATION (M/L**3)
C      VUNITS    UNITS OF GROUND-WATER VELOCITY (L/T)
C      DUNITS    UNITS OF DISPERSION COEFFICIENT (L**2/T)
C      KUNITS    UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C      LUNITS    UNITS OF LENGTH (L)
C      TUNITS    UNITS OF TIME (T)
C
C        DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
       CALL OFILE
       CALL TITLE
       WRITE(IO,201)
C
C        READ IN MODEL PARAMETERS
       READ(IN,101) NX,NY,NT,NMAX,IPLT
       WRITE(IO,205) NX,NY,NT,NMAX
       READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
       READ(IN,110) C0,VX,DX,DY,DK
       WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DK,KUNITS
       READ(IN,110) Y1,Y2
       WRITE(IO,212) Y1,LUNITS,Y2,LUNITS
       READ(IN,110) (X(I),I=1,NX)
       WRITE(IO,215) LUNITS
       WRITE(IO,220) (X(I),I=1,NX)
       READ(IN,110) (Y(I),I=1,NY)
       WRITE(IO,216) LUNITS
       WRITE(IO,220) (Y(I),I=1,NY)
       READ(IN,110) (T(I),I=1,NT)
       WRITE(IO,225) TUNITS
       WRITE(IO,220) (T(I),I=1,NT)
       IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
       IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C        READ IN GAUSS-LEGENDRE POINTS AND WEIGHTING FACTORS
       CALL GLQPTS (NMAX)
C
C        BEGIN TIME LOOP
       DO 20 IT=1,NT
C
C        BEGIN X LOOP
       DO 40 IX=1,NX
C
C      CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX)
```

```
      DO 50 IY=1,NY
      CALL CNRMLI(DK,T(IT),X(IX),Y(IY),Y1,Y2,DX,DY,VX,CN,NMAX)
      CXY(IX,IY)=CO*CN
 50   CONTINUE
 40   CONTINUE
C
C        PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
 70   IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
 60   CONTINUE
C
C        CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C        PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY CO TO PLOT
C        CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 20
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
      CP(IP+J)=SNGL(CXY(I,J)/CO)
 80   CONTINUE
      TP=SNGL(T(IT))
      NXY2=NXY*2
      CALL PLOT2D (XP,YP,CP,TP,DELTA,NX,NY,NXY,NXY2,IT,NT,IPLT,TUNITS,
     1 LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
 20   CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C        FORMAT STATEMENTS
 101  FORMAT(20I4)
 105  FORMAT(8A10)
 110  FORMAT(8F10.0)
 201  FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE TWO-DIMENSIONAL'/
     1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,29X,'FOR A SEMI-INFINITE AQUIFER OF INFINITE WIDTH'/
```

```
     3 1H ,26X,'WITH A FINITE-WIDTH (STRIP) SOLUTE SOURCE AT X=0.0'
     4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205   FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
     2 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
     3 'NUMBER OF POINTS FOR NUMERICAL INTEGRATION (NMAX) = ',I4)
210   FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (CO) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
     5 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212   FORMAT(1H0,25X,'AQUIFER WIDTH (W) IS INFINITE'/1H ,25X,
     1 'SOLUTE SOURCE IS LOCATED BETWEEN Y1 =',1P1E13.6,1X,A10/1H ,54X,
     2 'AND Y2 =',1P1E13.6,1X,A10)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216   FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227   FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
     1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
     2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10,5X,'(CONTINUED)'/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
235   FORMAT(1H ,20X,9F12.4)
236   FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240   FORMAT(1H ,5X,F12.4,2X,'!',9F12.5)
241   FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRMLI(DK,T,X,Y,Y1,Y2,DX,DY,VX,CN,NMAX)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      COMMON /IOUNIT/ IN,IO
      COMMON /GLPTS/ WN(256),ZN(256)
C
C        THIS ROUTINE CALCULATES THE NORMALIZED CONCENTRATION AT X,Y
C        BASED ON THE ANALYTIC SOLUTION TO THE TWO-DIMENSIONAL
C        ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION FOR A SEMI-
C        INFINITE AQUIFER OF INFINITE WIDTH.  A FINITE-WIDTH (STRIP)
C        SOLUTE SOURCE EXTENDS FROM Y=Y1 TO Y=Y2. THE SOLUTION CONTAINS
C        AN INTEGRAL FROM 0 TO T**.25 WHICH IS EVALUATED USING A GAUSS-
C        LEGENDRE QUADRATURE NUMERICAL INTEGRATION TECHNIQUE.
C
      PI=3.14159265358979D0
      CN=0.0D0
```

```
C
C         FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
        IF(T.LE.0.0D0) RETURN
C
C         FOR X=0.0, CONCENTRATIONS ARE SPECIFIED BY BOUNDARY CONDITIONS
        IF(X.GT.0.0D0) GO TO 10
        IF(Y.GT.Y1 .AND. Y.LT.Y2) CN=1.0D0
        IF(Y.EQ.Y1) CN=0.50D0
        IF(Y.EQ.Y2) CN=0.50D0
        RETURN
C
C         START NUMERICAL INTEGRATION LOOP
10      SUM=0.0D0
        DO 20 I=1,NMAX
C
C         SCALE THE GAUSS-LEGENDRE COEFFICIENTS TO ACCOUNT FOR THE
C         NON-NORMALIZED LIMITS OF INTEGRATION
          LIMITS OF INTEGRATION ARE FROM 0 TO T**0.25
        TT=T**0.250D0
        WI=WN(I)
        ZI=TT*(ZN(I)+1.0D0)/2.0D0
        ZSQ=ZI*ZI
        Z4=ZSQ*ZSQ
C
C         TERM 1
        XVT=X-VX*Z4
        EXP1=-XVT*XVT/(4.0D0*DX*Z4)-DK*Z4
        ERFC1=(Y1-Y)/(2.0D0*ZSQ*DSQRT(DY))
        CALL EXERFC(EXP1,ERFC1,Z1)
C
C         TERM 2
        ERFC2=(Y2-Y)/(2.0D0*ZSQ*DSQRT(DY))
        CALL EXERFC(EXP1,ERFC2,Z2)
        TERM=(Z1-Z2)*WI/(ZI*ZSQ)
        SUM=SUM+TERM
20      CONTINUE
        SUM=SUM*TT/2.0D0
        CN=SUM*X/DSQRT(PI*DX))
        RETURN
        END
```

```
C
C        ********************************************************
C        *                                                      *
C        *                   **** GAUSS ****                    *
C        *                                                      *
C        * TWO-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT MODEL  *
C        *                                                      *
C        *   FOR A SEMI-INFINITE AQUIFER OF INFINITE WIDTH. A   *
C        *                                                      *
C        *  SOURCE HAVING A GAUSSIAN-SHAPED CONCENTRATION DIS-  *
C        *                                                      *
C        * TRIBUTION IS LOCATED AT X=0 AND CENTERED ABOUT Y=YC  *
C        *                                                      *
C        *        GROUND-WATER FLOW IN X-DIRECTION ONLY         *
C        *                                                      *
C        *            VERSION CURRENT AS OF 04/01/90            *
C        *                                                      *
C        ********************************************************
C
C      THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C      GREATER THAN THOSE GIVEN HERE.
C        MAXX = MAXIMUM NUMBER OF X-VALUES
C        MAXY = MAXIMUM NUMBER OF Y-VALUES
C        MAXT = MAXIMUM NUMBER OF TIME VALUES
C        MAXXY = MAXX * MAXY
C        MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      REAL XP,YP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CM        MAXIMUM SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L*
C     DX        LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY        TRANSVERSE DISPERSION COEFFICIENT [L**2/T]
C     VX        GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK        FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X         X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y         Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T         TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN        NORMALIZED CONCENTRATION C/CM [DIMENSIONLESS]
C     CXY       SOLUTE CONCENTRATION C(X,Y,T) [M/L**3]
C     YC        Y-COORDINATE OF THE CENTER OF SOLUTE SOURCE AT X=0 [L]
```

```
C     SIGMA     STANDARD DEVIATION OF GAUSSIAN CONCENTRATION DISTRIBUTION
C               FOR THE SOLUTE SOURCE [L]
C
C     NX        NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NY        NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT        NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     NMAX      NUMBER OF TERMS USED IN GAUSS-LEGENDRE NUMERICAL
C               INTEGRATION TECHNIQUE (MUST EQUAL 4, 20, 60, 104 OR 256)
C
C     IPLT      PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C     XSCLP     SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C     YSCLP     SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C     DELTA     CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS    UNITS OF CONCENTRATION (M/L**3)
C     VUNITS    UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS    UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS    UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS    UNITS OF LENGTH (L)
C     TUNITS    UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NX,NY,NT,NMAX,IPLT
      WRITE(IO,205) NX,NY,NT,NMAX
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) CM,VX,DX,DY,DK
      WRITE(IO,210) CM,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DK,KUNITS
      READ(IN,110) YC,SIGMA
      WRITE(IO,212) YC,LUNITS,SIGMA,LUNITS
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (Y(I),I=1,NY)
      WRITE(IO,216) LUNITS
      WRITE(IO,220) (Y(I),I=1,NY)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
      IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
      IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C       READ IN GAUSS-LEGENDRE POINTS AND WEIGHTING FACTORS
      CALL GLQPTS (NMAX)
C
C       BEGIN TIME LOOP
      DO 20 IT=1,NT
C
```

```
C         BEGIN X LOOP
      DO 40 IX=1,NX
C
C     CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX)
      DO 50 IY=1,NY
      CALL CNRMLG(DK,T(IT),X(IX),Y(IY),YC,SIGMA,DX,DY,VX,CN,NMAX)
      CXY(IX,IY)=CM*CN
50    CONTINUE
40    CONTINUE
C
C       PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
70    IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
60    CONTINUE
C
C       CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C       PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY CM TO PLOT
C       CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 20
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
      CP(IP+J)=SNGL(CXY(I,J)/CM)
80    CONTINUE
      TP=SNGL(T(IT))
      NXY2=NXY*2
      CALL PLOT2D (XP,YP,CP,TP,DELTA,NX,NY,NXY,NXY2,IT,NT,IPLT,TUNITS,
     1 LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
20    CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C          FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
```

```
110    FORMAT(8F10.0)
201    FORMAT(/////1H ,30X,'ANALYTICAL SOLUTION TO THE TWO-DIMENSIONAL'/
      1 1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
      2 1H ,29X,'FOR A SEMI-INFINITE AQUIFER OF INFINITE WIDTH'/
      3 1H ,25X,'WITH A SOLUTE SOURCE HAVING A GAUSSIAN CONCENTRATION'/
      4 1H ,24X,'DISTRIBUTION LOCATED AT X=0.0 AND CENTERED ABOUT Y=YC'
      5 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205    FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
      1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
      2 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
      3 'NUMBER OF POINTS FOR NUMERICAL INTEGRATION (NMAX) = ',I4)
210    FORMAT(1H0,25X,'MAXIMUM SOLUTE CONCENTRATION AT THE BOUNDARY',
      1 ' CM) =',1P1E13.6,1X,A10/1H ,25X,
      2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
      3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
      4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
      5 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212    FORMAT(1H0,25X,'AQUIFER WIDTH (W) IS INFINITE'/1H ,25X,
      1 'SOLUTE SOURCE IS CENTERED AT Y =',1P1E13.6,1X,A10/1H ,25X,
      2 'STANDARD DEVIATION OF GAUSSIAN DISTRIBUTION (SIGMA) =',
      3 1P1E13.6,1X,A10)
215    FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216    FORMAT(1H0,25X,'Y COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220    FORMAT(1H ,5X,8F12.4)
225    FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
      1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227    FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
      1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
      2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230    FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
      1 F12.4,1X,A10/
      2 1H0,25X,'Y-COORDINATE, IN ',A10)
231    FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
      1 F12.4,1X,A10,5X,'(CONTINUED)'/
      2 1H0,25X,'Y-COORDINATE, IN ',A10)
235    FORMAT(1H ,20X,9F12.4)
236    FORMAT(1H ,19X,'*',108(1H-)/
      1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
      2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240    FORMAT(1H ,5X,F12.4,2X,'!',9F12.6)
241    FORMAT(1H ,19X,'!')
       END
       SUBROUTINE CNRMLG(DK,T,X,Y,YC,SIGMA,DX,DY,VX,CN,NMAX)
       IMPLICIT DOUBLE PRECISION(A-H,O-Z)
       COMMON /IOUNIT/ IN,IO
       COMMON /GLPTS/ WN(256),ZN(256)
C
C        THIS ROUTINE CALCULATES THE NORMALIZED CONCENTRATION AT X,Y
C        BASED ON THE ANALYTIC SOLUTION TO THE TWO-DIMENSIONAL
C        ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION FOR A SEMI-
C        INFINITE AQUIFER OF INFINITE WIDTH. THE SOLUTE SOURCE, LOCATED
```

```
C         AT X=0.0 AND CENTERED ABOUT Y=YC, HAS A GAUSSIAN CONCENTRATION
C         DISTRIBUTION WITH A STANDARD DEVIATION OF SIGMA. THE SOLUTION
C         CONTAINS AN INTEGRAL FROM 0 TO T**.25 WHICH IS EVALUATED
C         USING A GAUSS-LEGENDRE QUADRATURE INTEGRATION TECHNIQUE.
C
      PI=3.14159265358979D0
      Y1=Y-YC
      SIGSQ=SIGMA*SIGMA
      BETA=DK+VX*VX/(4.0D0*DX)
      VX2D=VX*X/(2.0D0*DX)
      CN=0.0D0
C
C         FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C         FOR X=0.0, CONCENTRATIONS ARE SPECIFIED BY BOUNDARY CONDITIONS
      IF(X.GT.0.0D0) GOTO 10
      CN=DEXP(-Y1*Y1/(2.0D0*SIGSQ))
      RETURN
C
C         START NUMERICAL INTEGRATION LOOP
10    SUM=0.0D0
      DO 20 I=1,NMAX
C
C         SCALE THE GAUSS-LEGENDRE COEFFICIENTS TO ACCOUNT FOR THE
C         NON-NORMALIZED LIMITS OF INTEGRATION
C         LIMITS OF INTEGRATION ARE FROM 0 TO T**0.25
      TT=T**0.250D0
      WI=WN(I)
      ZI=TT*(ZN(I)+1.0D0)/2.0D0
C
C         TERM 1
      Z4=ZI**4
      ALPHA=Z4*DY + SIGSQ/2.0D0
      X1=DEXP(-Z4*BETA -X*X/(4.0D0*DX*Z4) -Y1*Y1/(4.0D0*ALPHA))
      X1=X1/((ZI*ZI*ZI)*DSQRT(ALPHA))
      SUM=SUM+X1*WI
20    CONTINUE
      SUM=SUM*TT/2.0D0
C
C         TERM 2
      X2=SUM*2.0D0*DEXP(VX2D)
      CN=X*SIGMA*X2/(DSQRT(2.0D0*PI*DX))
      RETURN
      END
```

```
C
C       ********************************************************
C       *                                                      *
C       *                  **** POINT3 ****                    *
C       *                                                      *
C       *   THREE-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT    *
C       *                                                      *
C       *    MODEL FOR AN AQUIFER OF INFINITE EXTENT WITH A    *
C       *                                                      *
C       *   CONTINUOUS POINT SOURCE AT X=XC, Y=YC, AND Z=ZC    *
C       *                                                      *
C       *        GROUND-WATER FLOW IN X-DIRECTION ONLY         *
C       *                                                      *
C       *            VERSION CURRENT AS OF 04/01/90            *
C       *                                                      *
C       ********************************************************
C
C     THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C     GREATER THAN THOSE GIVEN HERE.
C       MAXX = MAXIMUM NUMBER OF X-VALUES
C       MAXY = MAXIMUM NUMBER OF Y-VALUES
C       MAXZ = MAXIMUM NUMBER OF Z-VALUES
C       MAXT = MAXIMUM NUMBER OF TIME VALUES
C       MAXXY = MAXX * MAXY
C       MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXZ=30,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,QUNITS,TUNITS
      REAL XP,YP,ZP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),Z(MAXZ),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C         NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C         MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C         LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO       SOLUTE CONCENTRATION IN INJECTED FLUID [M/L**3]
C     DX       LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY       TRANSVERSE (Y-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     DZ       TRANSVERSE (Z-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     VX       GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK       FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X        X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y        Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Z        Z-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T        TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN       NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXY      SOLUTE CONCENTRATION C(X,Y,Z,T) [M/L**3]
```

```
C     XC        X-COORDINATE OF CONTINUOUS POINT SOURCE [L]
C     YC        Y-COORDINATE OF CONTINUOUS POINT SOURCE [L]
C     ZC        Z-COORDINATE OF CONTINUOUS POINT SOURCE [L]
C     QM        FLUID INJECTION RATE [L**3/T]
C     POR       AQUIFER POROSITY [DIMENSIONLESS]
C
C     NX        NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NY        NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NZ        NUMBER OF Z-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT        NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C
C     IPLT      PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C     XSCLP     SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C     YSCLP     SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C     DELTA     CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS    UNITS OF CONCENTRATION (M/L**3)
C     VUNITS    UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS    UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS    UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS    UNITS OF LENGTH (L)
C     QUNITS    UNITS OF FLUID INJECTION RATE (L**3/T)
C     TUNITS    UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NX,NY,NZ,NT,IPLT
      WRITE(IO,205) NX,NY,NZ,NT
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,QUNITS,TUNITS
      READ(IN,110) C0,VX,DX,DY,DZ,DK
      WRITE(IO,210) C0,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DZ,DUNITS,
     1 DK,KUNITS
      READ(IN,110) XC,YC,ZC,QM,POR
      WRITE(IO,212) XC,LUNITS,YC,LUNITS,ZC,LUNITS,QM,QUNITS,POR
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (Y(I),I=1,NY)
      WRITE(IO,216) LUNITS
      WRITE(IO,220) (Y(I),I=1,NY)
      READ(IN,110) (Z(I),I=1,NZ)
      WRITE(IO,217) LUNITS
      WRITE(IO,220) (Z(I),I=1,NZ)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
      IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
      IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
```

```
C
C       BEGIN TIME LOOP
      DO 20 IT=1,NT
C
C       BEGIN Z LOOP
      DO 30 IZ=1,NZ
      ZZ=Z(IZ)-ZC
C
C       BEGIN X LOOP
      DO 40 IX=1,NX
      XX=X(IX)-XC
C
C     CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX) AND Z=Z(IZ
      DO 50 IY=1,NY
      YY=Y(IY)-YC
      CALL CNRML3(QM,POR,DK,T(IT),XX,YY,ZZ,DX,DY,DZ,VX,CN)
      CXY(IX,IY)=C0*CN
50    CONTINUE
40    CONTINUE
C
C       PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
70    IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
60    CONTINUE
C
C       CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C       PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY C0 TO PLOT
C       CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 30
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
      CP(IP+J)=SNGL(CXY(I,J)/C0)
80    CONTINUE
      TP=SNGL(T(IT))
      ZP=SNGL(Z(IZ))
```

```
      NXY2=NXY*2
      CALL PLOT3D (XP,YP,ZP,CP,TP,DELTA,NX,NY,NXY,NXY2,IZ,NZ,IPLT,
     1 TUNITS,LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
30    CONTINUE
20    CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C         FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
110   FORMAT(8F10.0)
201   FORMAT(/////1H ,29X,'ANALYTICAL SOLUTION TO THE THREE-DIMENSIONAL'
     1 /1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,34X,'FOR AN AQUIFER OF INFINITE EXTENT'/
     3 1H ,30X,'WITH A CONTINUOUS POINT SOURCE AT XC,YC,ZC'
     4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205   FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
     2 'NUMBER OF Z-COORDINATES (NZ) = ',I4/1H ,25X,
     3 'NUMBER OF TIME VALUES (NT) = ',I4)
210   FORMAT(1H0,25X,'SOLUTE CONCENTRATION IN INJECTED FLUID (C0) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
     5 1H ,25X,'DISPERSION IN THE Z-DIRECTION (DZ) =',1P1E13.6,1X,A10/
     6 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212   FORMAT(1H0,25X,'AQUIFER IS OF INFINITE EXTENT'
     2 /1H0,25X,'CONTINUOUS POINT SOURCE IS AT X =',1P1E13.6,1X,A10/
     3 1H ,55X,'Y =',1P1E13.6,1X,A10/1H ,55X,'Z =',1P1E13.6,1X,A10/
     5 1H ,25X,'FLUID INJECTION RATE (QM) =',1P1E13.6,1X,A10/
     6 1H ,25X,'AQUIFER POROSITY (POR) =',1P1E13.6)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216   FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
217   FORMAT(1H0,25X,'Z-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227   FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
     1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
     2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10/1H ,35X,'AND AT Z =',F12.4,1X,A10/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10,5X,'(CONTINUED)'/1H ,35X,'AND AT Z =',F12.4,1X,A10/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
235   FORMAT(1H ,20X,9F12.4)
```

```
236   FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240   FORMAT(1H ,5X,F12.4,2X,'!',9F12.6)
241   FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRML3(QM,POR,DK,T,X,Y,Z,DX,DY,DZ,VX,CN)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      COMMON /IOUNIT/ IN,IO
C
C       THIS ROUTINE CALCULATES SOLUTE CONCENTRATION AT X,Y,Z BASED ON
C       THE ANALYTIC SOLUTION TO THE THREE-DIMENSIONAL ADVECTIVE-
C       DISPERSIVE SOLUTE TRANSPORT EQUATION FOR AN AQUIFER OF
C       INFINITE EXTENT WITH A CONTINUOUS POINT SOURCE LOCATED AT
C       X=XC, Y=YC, AND Z=ZC.  A CLOSED FORM SOLUTION WAS OBTAINED.
C
      PI=3.14159265358979D0
      CN=0.0D0
C
C       FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C        CHECK FOR X=Y=Z=0
      IF(X.EQ.0.0D0 .AND. Y.EQ.0.0D0 .AND. Z.EQ.0.0D0) THEN
        WRITE(IO,200)
        RETURN
      END IF
C
      BETA=DSQRT(VX*VX+4.0D0*DX*DK)
      GAMMA=DSQRT(X*X+Y*Y*DX/DY+Z*Z*DX/DZ)
      RTDXT=2.0D0*DSQRT(DX*T)
C
C       TERM 1
      X1=(VX*X-GAMMA*BETA)/(2.0D0*DX)
      Y1=(GAMMA-BETA*T)/RTDXT
      CALL EXERFC(X1,Y1,Z1)
C
C       TERM 2
      X2=(VX*X+GAMMA*BETA)/(2.0D0*DX)
      Y2=(GAMMA+BETA*T)/RTDXT
      CALL EXERFC(X2,Y2,Z2)
C
C       TERM 3
      Z3=Z1+Z2
      Z4=DSQRT(DY*DZ)
      CN=QM*Z3/(8.0D0*POR*PI*GAMMA*Z4)
      RETURN
C
C       FORMAT STATEMENTS
200   FORMAT (1H0,5X,'**** WARNING ****  A SOLUTION CAN NOT BE COMPUTED'
     1 ' FOR X=XC,Y=YC,Z=ZC'/)
      END
```

```
C
C          ******************************************************
C          *                                                    *
C          *                  **** PATCHF ****                  *
C          *                                                    *
C          *   THREE-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT  *
C          *                                                    *
C          *   MODEL FOR A SEMI-INFINITE AQUIFER WITH A FINITE  *
C          *                                                    *
C          *    WIDTH AND HEIGHT. A PATCH SOURCE EXTENDS FROM   *
C          *                                                    *
C          *             Y1 TO Y2 AND Z1 TO Z2 AT X=0           *
C          *                                                    *
C          *         GROUND-WATER FLOW IN X-DIRECTION ONLY      *
C          *                                                    *
C          *            VERSION CURRENT AS OF 04/01/90          *
C          *                                                    *
C          ******************************************************
C
C       THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C       GREATER THAN THOSE GIVEN HERE.
C         MAXX = MAXIMUM NUMBER OF X-VALUES
C         MAXY = MAXIMUM NUMBER OF Y-VALUES
C         MAXZ = MAXIMUM NUMBER OF Z-VALUES
C         MAXT = MAXIMUM NUMBER OF TIME VALUES
C         MAXXY = MAXX * MAXY
C         MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXZ=30,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      CHARACTER*1 IERR(MAXX,MAXY)
      REAL XP,YP,ZP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),Z(MAXZ),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C           NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C           MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C           LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO        SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX        LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY        TRANSVERSE (Y-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     DZ        TRANSVERSE (Z-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     VX        GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK        FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X         X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y         Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Z         Z-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
```

```
C     T          TIME AT WHICH CONCENTRATION IS EVALUATED [T]
C     CN         NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXY        SOLUTE CONCENTRATION C(X,Y,Z,T) [M/L**3]
C     W          AQUIFER WIDTH (AQUIFER EXTENDS FROM Y=0 TO Y=W) [L]
C     H          AQUIFER HEIGHT (AQUIFER EXTENDS FROM Z=0 TO Z=H) [L]
C     Y1         Y-COORDINATE OF LOWER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Y2         Y-COORDINATE OF UPPER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Z1         Z-COORDINATE OF LOWER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Z2         Z-COORDINATE OF UPPER LIMIT OF PATCH SOLUTE SOURCE [L]
C
C     NX         NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NY         NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NZ         NUMBER OF Z-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT         NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     NMAX       NUMBER OF TERMS USED IN INNER INFINITE SERIES SUMMATION
C     MMAX       NUMBER OF TERMS USED IN OUTER INFINITE SERIES SUMMATION
C
C     IPLT       PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C     XSCLP      SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C     YSCLP      SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C     DELTA      CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS   UNITS OF CONCENTRATION (M/L**3)
C     VUNITS   UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS   UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS   UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS   UNITS OF LENGTH (L)
C     TUNITS   UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NX,NY,NZ,NT,NMAX,MMAX,IPLT
      WRITE(IO,205) NX,NY,NZ,NT,NMAX,MMAX
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) CO,VX,DX,DY,DZ,DK
      WRITE(IO,210) CO,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DZ,DUNITS,
     1 DK,KUNITS
      READ(IN,110) W,H,Y1,Y2,Z1,Z2
      WRITE(IO,212) W,LUNITS,H,LUNITS,Y1,LUNITS,Y2,LUNITS,Z1,LUNITS,
     1 Z2,LUNITS
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (Y(I),I=1,NY)
      WRITE(IO,216) LUNITS
      WRITE(IO,220) (Y(I),I=1,NY)
      READ(IN,110) (Z(I),I=1,NZ)
      WRITE(IO,217) LUNITS
```

```
      WRITE(IO,220) (Z(I),I=1,NZ)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
      IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
      IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C        BEGIN TIME LOOP
      DO 20 IT=1,NT
C
C        BEGIN Z LOOP
      DO 30 IZ=1,NZ
C
C        BEGIN X LOOP
      DO 40 IX=1,NX
C
C     CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX) AND Z=Z(IZ
      DO 50 IY=1,NY
      CALL CNRMLP(DK,T(IT),X(IX),Y(IY),Z(IZ),W,H,Y1,Y2,Z1,Z2,DX,
     1 DY,DZ,VX,CN,NMAX,MMAX,IERR(IX,IY))
      CXY(IX,IY)=CO*CN
50    CONTINUE
40    CONTINUE
C
C        PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),IERR(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
70    IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
60    CONTINUE
C
C        CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C        PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY CO TO PLOT
C        CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 30
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
```

```
      CP(IP+J)=SNGL(CXY(I,J)/C0)
 80   CONTINUE
      TP=SNGL(T(IT))
      ZP=SNGL(Z(IZ))
      NXY2=NXY*2
      CALL PLOT3D (XP,YP,ZP,CP,TP,DELTA,NX,NY,NXY,NXY2,IZ,NZ,IPLT,
     1 TUNITS,LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
 30   CONTINUE
 20   CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C          FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
110   FORMAT(8F10.0)
201   FORMAT(/////1H ,29X,'ANALYTICAL SOLUTION TO THE THREE-DIMENSIONAL'
     1 /1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,30X,'FOR A SEMI-INFINITE AQUIFER OF FINITE WIDTH'/
     3 1H ,28X,'AND HEIGHT WITH A PATCH SOLUTE SOURCE AT X=0.0'
     4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205   FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
     2 'NUMBER OF Z-COORDINATES (NZ) = ',I4/1H ,25X,
     3 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
     4 'NUMBER OF TERMS IN INNER INFINTE SERIES SUMMATION (NMAX) = ',
     5 I4/1H ,25X,
     6 'NUMBER OF TERMS IN OUTER INFINTE SERIES SUMMATION (MMAX) = ',I4)
210   FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (C0) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
     5 1H ,25X,'DISPERSION IN THE Z-DIRECTION (DZ) =',1P1E13.6,1X,A10/
     6 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212   FORMAT(1H0,25X,'AQUIFER WIDTH (W) =',1P1E13.6,1X,A10/1H ,25X,
     1 'AQUIFER HEIGHT (H) =',1P1E13.6,1X,A10/1H ,25X,
     2 'SOLUTE SOURCE IS LOCATED BETWEEN Y1 =',1P1E13.6,1X,A10/1H ,58X,
     3 'Y2 =',1P1E13.6,1X,A10/1H ,58X,
     4 'Z1 =',1P1E13.6,1X,A10/1H ,54X,
     5 'AND Z2 =',1P1E13.6,1X,A10)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216   FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
217   FORMAT(1H0,25X,'Z-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227   FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
     1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
```

```
     2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10/1H ,35X,'AND AT Z =',F12.4,1X,A10,
     1 15X,'*  INDICATES SOLUTION DID NOT CONVERGE'/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10,5X,'(CONTINUED)'/1H ,35X,'AND AT Z =',F12.4,1X,
     2 A10/1H0,25X,'Y-COORDINATE, IN ',A10)
235   FORMAT(1H ,20X,9F12.4)
236   FORMAT(1H ,19X,'*',108(1H-)/
     1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
     2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
240   FORMAT(1H ,5X,F12.4,2X,'! ',9(F11.5,A1))
241   FORMAT(1H ,19X,'!')
      END
      SUBROUTINE CNRMLP(DK,T,X,Y,Z,W,H,Y1,Y2,Z1,Z2,DX,DY,DZ,VX,CN,NMAX,
     1 MMAX,IERR)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      CHARACTER*1 IERR
      COMMON /IOUNIT/ IN,IO
C
C       THIS ROUTINE CALCULATES THE NORMALIZED CONCENTRATION AT X,Y,Z
C       BASED ON THE ANALYTIC SOLUTION TO THE THREE-DIMENSIONAL
C       ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION FOR A SEMI-
C       INFINITE AQUIFER WITH A FINITE WIDTH AND HEIGHT. THE SOLUTE
C       SOURCE HAS A FINITE WIDTH AND HEIGHT, EXTENDING FROM Y=Y1 TO
C       Y=Y2 AND Z=Z1 TO Z=Z2. SOLUTE MAY BE SUBJECT TO FIRST-ORDER
C       CHEMICAL TRANSFORMATION. THE SOLUTION CONTAINS TWO INFINITE
C       SERIES SUMMATIONS WHICH MAY CONVERGE SLOWLY.
C
      PI=3.14159265358979D0
      CN=0.0D0
      IERR=' '
C
C       FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C       FOR X=0.0, CONCENTRATIONS ARE SPECIFIED BY BOUNDARY CONDITIONS
      IF(X.GT.0.0D0) GO TO 10
      IF(Y.EQ.Y1.OR.Y.EQ.Y2) THEN
        IF(Z.GT.Z1.AND.Z.LT.Z2) CN=0.50D0
        IF(Z.EQ.Z1.OR.Z.EQ.Z2) CN=0.25D0
      END IF
      IF(Z.EQ.Z1.OR.Z.EQ.Z2) THEN
        IF(Y.GT.Y1.AND.Y.LT.Y2) CN=0.50D0
      END IF
      IF(Y.GT.Y1.AND.Y.LT.Y2.AND.Z.GT.Z1.AND.Z.LT.Z2) CN=1.0D0
      RETURN
10    RTDXT=2.0D0*DSQRT(DX*T)
C
C       BEGIN SUMMATION OF TERMS IN INFINITE SERIES (OUTER SERIES)
      NMAX1=NMAX+1
      MMAX1=MMAX+1
```

```
      SIGMAM=0.0D0
      SUBTM=0.0D0
      DO 20 MM=1,MMAX1
      M=MM-1
      ZETA=M*PI/H
      OM=(Z2-Z1)/H
      IF(M.NE.0) OM=(DSIN(ZETA*Z2)-DSIN(ZETA*Z1))/(M*PI)
      COSSZ=DCOS(ZETA*Z)
C
C       BEGIN SUMMATION OF TERMS IN INFINITE SERIES (INNER SERIES)
      SIGMAN=0.0D0
      SUBTN=0.0D0
      DO 30 NN=1,NMAX1
      N=NN-1
      ETA=N*PI/W
      PN=(Y2-Y1)/W
      IF(N.NE.0) PN=(DSIN(ETA*Y2)-DSIN(ETA*Y1))/(N*PI)
      COSRY=DCOS(ETA*Y)
      ALPHA=4.0D0*DX*(ETA*ETA*DY+ZETA*ZETA*DZ+DK)
      BETA=DSQRT(VX*VX+ALPHA)
      BETAT=BETA*T
C
C     IF M>0 AND N>0, USE GENERAL FORM
C       TERM 1
      A1=X*(VX-BETA)/(2.0D0*DX)
      B1=(X-BETAT)/RTDXT
      CALL EXERFC(A1,B1,C1)
      A2=X*(VX+BETA)/(2.0D0*DX)
      B2=(X+BETAT)/RTDXT
      CALL EXERFC(A2,B2,C2)
      TERM1=COSRY*PN*(C1+C2)
C
C       MULTIPLY TERM BY L(MN)
      IF(M.EQ.0 .AND. N.EQ.0) TERM1=TERM1*0.50D0
      IF(M.GT.0 .AND. N.GT.0) TERM1=TERM1*2.0D0
C
C       ADD TERM TO SUMMATION
      SIGMAN=SIGMAN+TERM1
C
C       CHECK FOR CONVERGENCE OF INNER SERIES.  BECAUSE SERIES
C       OSCILLATES, CHECK SUBTOTAL OF LAST 10 TERMS.
      SUBTN=SUBTN+TERM1
      IF(MOD(NN,10).NE.0) GO TO 30
      IF(DABS(SUBTN).LT.1.0D-12) GO TO 25
      SUBTN=0.0D0
30    CONTINUE
      IERR='*'
25    SIGMAM=SIGMAM+SIGMAN*COSSZ*OM
C
C       CHECK FOR CONVERGENCE OF OUTER SERIES.  BECAUSE SERIES
C       OSCILLATES, CHECK SUBTOTAL OF LAST 10 TERMS.
      SUBTM=SUBTM+SIGMAN*OM*COSSZ
      IF(MOD(MM,10).NE.0) GO TO 20
```

```
      IF(DABS(SUBTM).LT.1.0D-12) GO TO 35
      SUBTM=0.0D0
20    CONTINUE
      IERR='*'
35    CN=SIGMAM
      RETURN
      END
```

```
C
C        ********************************************************
C        *                                                      *
C        *                  **** PATCHI ****                    *
C        *                                                      *
C        *  THREE-DIMENSIONAL GROUND-WATER SOLUTE-TRANSPORT     *
C        *                                                      *
C        *   MODEL FOR A SEMI-INFINITE AQUIFER OF INFINITE      *
C        *                                                      *
C        *   WIDTH AND HEIGHT. PATCH SOURCE EXTENDING FROM      *
C        *                                                      *
C        *        Y1 TO Y2 AND Z1 TO Z2 LOCATED AT X=0          *
C        *                                                      *
C        *       GROUND-WATER FLOW IN X-DIRECTION ONLY          *
C        *                                                      *
C        *           VERSION CURRENT AS OF 04/01/90             *
C        *                                                      *
C        ********************************************************
C
C     THE FOLLOWING CARD MUST BE CHANGED IF PROBLEM DIMENSIONS ARE
C     GREATER THAN THOSE GIVEN HERE.
C       MAXX = MAXIMUM NUMBER OF X-VALUES
C       MAXY = MAXIMUM NUMBER OF Y-VALUES
C       MAXZ = MAXIMUM NUMBER OF Z-VALUES
C       MAXT = MAXIMUM NUMBER OF TIME VALUES
C       MAXXY = MAXX * MAXY
C       MAXXY2 = 2 * MAXX * MAXY
      PARAMETER MAXX=100,MAXY=50,MAXZ=30,MAXT=20,MAXXY=5000,MAXXY2=10000
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      CHARACTER*10 CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      REAL XP,YP,ZP,CP,TP,DELTA,XPC,YPC,XSCLP,YSCLP
      DIMENSION CXY(MAXX,MAXY),X(MAXX),Y(MAXY),Z(MAXZ),T(MAXT)
      COMMON /PDAT/ XP(MAXX),YP(MAXY),CP(MAXXY),XPC(50),YPC(50),
     1 IFLAG(MAXXY2)
      COMMON /IOUNIT/ IN,IO
C
C     PROGRAM VARIABLES
C
C         NOTE:  ANY CONSISTANT SET OF UNITS MAY BE USED IN THE
C         MODEL.  NO FORMAT STATEMENTS NEED TO BE CHANGED AS
C         LABELS FOR ALL VARIABLES ARE SPECIFIED IN MODEL INPUT.
C
C     CO        SOLUTE CONCENTRATION AT THE INFLOW BOUNDARY [M/L**3]
C     DX        LONGITUDINAL DISPERSION COEFFICIENT [L**2/T]
C     DY        TRANSVERSE (Y-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     DZ        TRANSVERSE (Z-DIRECTION) DISPERSION COEFFICIENT [L**2/T]
C     VX        GROUND-WATER VELOCITY IN X-DIRECTION [L/T]
C     DK        FIRST-ORDER SOLUTE DECAY CONSTANT [1/T]
C     X         X-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Y         Y-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     Z         Z-POSITION AT WHICH CONCENTRATION IS EVALUATED [L]
C     T         TIME AT WHICH CONCENTRATION IS EVALUATED [T]
```

```
C     CN        NORMALIZED CONCENTRATION C/CO [DIMENSIONLESS]
C     CXY       SOLUTE CONCENTRATION C(X,Y,Z,T) [M/L**3]
C     Y1        Y-COORDINATE OF LOWER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Y2        Y-COORDINATE OF UPPER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Z1        Z-COORDINATE OF LOWER LIMIT OF PATCH SOLUTE SOURCE [L]
C     Z2        Z-COORDINATE OF UPPER LIMIT OF PATCH SOLUTE SOURCE [L]
C
C     NX        NUMBER OF X-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NY        NUMBER OF Y-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NZ        NUMBER OF Z-POSITIONS AT WHICH SOLUTION IS EVALUATED
C     NT        NUMBER OF TIME VALUES AT WHICH SOLUTION IS EVALUATED
C     NMAX      NUMBER OF TERMS USED IN GAUSS-LEGENDRE NUMERICAL
C               INTEGRATION TECHNIQUE (MUST EQUAL 4, 20, 60, 104 OR 256)
C
C     IPLT      PLOT CONTROL. IF IPLT>0, CONTOUR MAPS ARE PLOTTED
C     XSCLP     SCALING FACTOR TO CONVERT X TO PLOTTER INCHES
C     YSCLP     SCALING FACTOR TO CONVERT Y TO PLOTTER INCHES
C     DELTA     CONTOUR INCREMENT FOR PLOT. (VALUE BETWEEN 0 AND 1.0)
C
C       CHARACTER VARIABLES USED TO SPECIFY UNITS FOR MODEL PARAMETERS
C     CUNITS    UNITS OF CONCENTRATION (M/L**3)
C     VUNITS    UNITS OF GROUND-WATER VELOCITY (L/T)
C     DUNITS    UNITS OF DISPERSION COEFFICIENT (L**2/T)
C     KUNITS    UNITS OF SOLUTE DECAY CONSTANT  (1/T)
C     LUNITS    UNITS OF LENGTH (L)
C     TUNITS    UNITS OF TIME (T)
C
C       DEFINE INPUT/OUTPUT FILES AND PRINT TITLE PAGE
      CALL OFILE
      CALL TITLE
      WRITE(IO,201)
C
C       READ IN MODEL PARAMETERS
      READ(IN,101) NX,NY,NZ,NT,NMAX,IPLT
      WRITE(IO,205) NX,NY,NZ,NT,NMAX
      READ(IN,105) CUNITS,VUNITS,DUNITS,KUNITS,LUNITS,TUNITS
      READ(IN,110) CO,VX,DX,DY,DZ,DK
      WRITE(IO,210) CO,CUNITS,VX,VUNITS,DX,DUNITS,DY,DUNITS,DZ,DUNITS,
     1 DK,KUNITS
      READ(IN,110) Y1,Y2,Z1,Z2
      WRITE(IO,212) Y1,LUNITS,Y2,LUNITS,Z1,LUNITS,Z2,LUNITS
      READ(IN,110) (X(I),I=1,NX)
      WRITE(IO,215) LUNITS
      WRITE(IO,220) (X(I),I=1,NX)
      READ(IN,110) (Y(I),I=1,NY)
      WRITE(IO,216) LUNITS
      WRITE(IO,220) (Y(I),I=1,NY)
      READ(IN,110) (Z(I),I=1,NZ)
      WRITE(IO,217) LUNITS
      WRITE(IO,220) (Z(I),I=1,NZ)
      READ(IN,110) (T(I),I=1,NT)
      WRITE(IO,225) TUNITS
      WRITE(IO,220) (T(I),I=1,NT)
```

```
      IF(IPLT.GT.0) READ(IN,110) XSCLP,YSCLP,DELTA
      IF(IPLT.GT.0) WRITE(IO,227) XSCLP,YSCLP,DELTA,CUNITS
C
C       READ IN GAUSS-LEGENDRE POINTS AND WEIGHTING FACTORS
      CALL GLQPTS (NMAX)
C
C       BEGIN TIME LOOP
      DO 20 IT=1,NT
C
C       BEGIN Z LOOP
      DO 30 IZ=1,NZ
C
C       BEGIN X LOOP
      DO 40 IX=1,NX
C
C     CALCULATE NORMALIZED CONCENTRATION FOR ALL Y AT X=X(IX) AND Z=Z(IZ
      DO 50 IY=1,NY
      CALL CNRMLP(DK,T(IT),X(IX),Y(IY),Z(IZ),Y1,Y2,Z1,Z2,DX,
     1 DY,DZ,VX,CN,NMAX)
      CXY(IX,IY)=CO*CN
   50 CONTINUE
   40 CONTINUE
C
C       PRINT OUT TABLES OF CONCENTRATION VALUES
      NPAGE=1+(NY-1)/9
      DO 60 NP=1,NPAGE
      IF(NP.EQ.1) WRITE(IO,230) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      IF(NP.NE.1) WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      NP1=(NP-1)*9
      NP2=9
      IF((NP1+NP2).GT.NY) NP2=NY-NP1
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
      DO 70 IX=1,NX
      WRITE(IO,240) X(IX),(CXY(IX,NP1+J),J=1,NP2)
      IF(MOD(IX,45).NE.0) GO TO 70
      WRITE(IO,231) T(IT),TUNITS,Z(IZ),LUNITS,LUNITS
      WRITE(IO,235) (Y(NP1+J),J=1,NP2)
      WRITE(IO,236) CUNITS,LUNITS
   70 IF(MOD(IX,5).EQ.0 .AND. MOD(IX,45).NE.0) WRITE(IO,241)
   60 CONTINUE
C
C       CONVERT X AND Y TO SINGLE PRECISION AND DIVIDE BY THE
C       PLOT SCALING FACTORS. CONVERT C(X,Y) AND DIVIDE BY CO TO PLOT
C       CONTOUR MAPS OF NORMALIZED CONCENTRATION FOR EACH TIME VALUE.
      IF(IPLT.LT.1) GO TO 30
      NXY=NX*NY
      DO 80 I=1,NX
      IP=(I-1)*NY
      XP(I)=SNGL(X(I))
      DO 80 J=1,NY
      IF(I.EQ.1) YP(J)=SNGL(Y(J))
      CP(IP+J)=SNGL(CXY(I,J)/CO)
```

```
80    CONTINUE
      TP=SNGL(T(IT))
      ZP=SNGL(Z(IZ))
      NXY2=NXY*2
      CALL PLOT3D (XP,YP,ZP,CP,TP,DELTA,NX,NY,NXY,NXY2,IZ,NZ,IPLT,
     1 TUNITS,LUNITS,XSCLP,YSCLP,XPC,YPC,IFLAG)
30    CONTINUE
20    CONTINUE
      CLOSE (IN)
      CLOSE (IO)
      STOP
C
C        FORMAT STATEMENTS
101   FORMAT(20I4)
105   FORMAT(8A10)
110   FORMAT(8F10.0)
201   FORMAT(/////1H ,29X,'ANALYTICAL SOLUTION TO THE THREE-DIMENSIONAL'
     1 /1H ,28X,'ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION'/
     2 1H ,30X,'FOR A SEMI-INFINITE AQUIFER OF INFINITE WIDTH'/
     3 1H ,28X,'AND HEIGHT WITH A PATCH SOLUTE SOURCE AT X=0.0'
     4 ///1H0,40X,'INPUT DATA'/1H ,40X,10(1H-))
205   FORMAT(1H0,25X,'NUMBER OF X-COORDINATES (NX) = ',I4/1H ,25X,
     1 'NUMBER OF Y-COORDINATES (NY) = ',I4/1H ,25X,
     2 'NUMBER OF Z-COORDINATES (NZ) = ',I4/1H ,25X,
     3 'NUMBER OF TIME VALUES (NT) = ',I4/1H ,25X,
     4 'NUMBER OF POINTS FOR NUMERICAL INTEGRATION (NMAX) = ',I4)
210   FORMAT(1H0,25X,'SOLUTE CONCENTRATION ON MODEL BOUNDARY (C0) =',
     1 1P1E13.6,1X,A10/1H ,25X,
     2 'GROUND-WATER VELOCITY IN X-DIRECTION (VX) =',1P1E13.6,1X,A10/
     3 1H ,25X,'DISPERSION IN THE X-DIRECTION (DX) =',1P1E13.6,1X,A10/
     4 1H ,25X,'DISPERSION IN THE Y-DIRECTION (DY) =',1P1E13.6,1X,A10/
     5 1H ,25X,'DISPERSION IN THE Z-DIRECTION (DZ) =',1P1E13.6,1X,A10/
     6 1H ,25X,'FIRST-ORDER SOLUTE DECAY RATE (DK) =',1P1E13.6,1X,A10)
212   FORMAT(1H0,25X,'AQUIFER WIDTH (W) AND HEIGHT (H) ARE INFINITE'
     2 /1H ,25X,'SOLUTE SOURCE IS LOCATED BETWEEN Y1 =',1P1E13.6,1X,A10/
     3 1H ,58X,'Y2 =',1P1E13.6,1X,A10/1H ,58X,
     4 'Z1 =',1P1E13.6,1X,A10/1H ,54X,
     5 'AND Z2 =',1P1E13.6,1X,A10)
215   FORMAT(1H0,25X,'X-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
216   FORMAT(1H0,25X,'Y-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
217   FORMAT(1H0,25X,'Z-COORDINATES AT WHICH SOLUTE CONCENTRATIONS ',
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,78(1H-)/)
220   FORMAT(1H ,5X,8F12.4)
225   FORMAT(1H0,25X,'TIMES AT WHICH SOLUTE CONCENTRATIONS '
     1 'WILL BE CALCULATED, IN ',A10/1H ,25X,70(1H-)/)
227   FORMAT(1H0,25X,'PLOT SCALING FACTOR FOR X (XSCLP) =',1P1E13.6/
     1 1H ,25X,'PLOT SCALING FACTOR FOR Y (YSCLP) =',1P1E13.6/
     2 1H ,25X,'CONTOUR INCREMENT (DELTA) =',1P1E13.6,1X,A10)
230   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
     1 F12.4,1X,A10/1H ,35X,'AND AT Z =',F12.4,1X,A10/
     2 1H0,25X,'Y-COORDINATE, IN ',A10)
```

```
 231   FORMAT(1H1/1H0,15X,'SOLUTE CONCENTRATION AT TIME =',
      1 F12.4,1X,A10,5X,'(CONTINUED)'/1H ,35X,'AND AT Z =',F12.4,1X,A10/
      2 1H0,25X,'Y-COORDINATE, IN ',A10)
 235   FORMAT(1H ,20X,9F12.4)
 236   FORMAT(1H ,19X,'*',108(1H-)/
      1 1H ,4X,'X-COORDINATE,',2X,'!',44X,'SOLUTE CONCENTRATION, IN '
      2 A10/1H ,4X,'IN ',A10,2X,1H!/1H ,19X,'!')
 240   FORMAT(1H ,5X,F12.4,2X,'!',9F12.6)
 241   FORMAT(1H ,19X,'!')
       END
       SUBROUTINE CNRMLP(DK,T,X,Y,Z,Y1,Y2,Z1,Z2,DX,DY,DZ,VX,CN,NMAX)
       IMPLICIT DOUBLE PRECISION(A-H,O-Z)
       COMMON /IOUNIT/ IN,IO
       COMMON /GLPTS/ WN(256),ZN(256)
C
C        THIS ROUTINE CALCULATES THE NORMALIZED CONCENTRATION AT X,Y,Z
C        BASED ON THE ANALYTIC SOLUTION TO THE THREE-DIMENSIONAL
C        ADVECTIVE-DISPERSIVE SOLUTE TRANSPORT EQUATION FOR A SEMI-
C        INFINITE AQUIFER WITH INFINITE WIDTH AND HEIGHT. THE SOLUTE
C        SOURCE HAS A FINITE WIDTH AND HEIGHT, EXTENDING FROM Y=Y1 TO
C        Y=Y2 AND Z=Z1 TO Z=Z2. THE SOLUTE MAY BE SUBJECT TO FIRST-ORDER
C        CHEMICAL TRANSFORMATION. THE SOLUTION CONTAINS AN INTEGRAL
C        FROM 0 TO T**.25 WHICH IS EVALUATED NUMERICALLY USING A GAUSS-
C        LEGENDRE QUADRATURE TECHNIQUE.
C
      PI=3.14159265358979D0
      CN=0.0D0
C
C        FOR T=0, ALL CONCENTRATIONS EQUAL 0.0
      IF(T.LE.0.0D0) RETURN
C
C        FOR X=0.0, CONCENTRATIONS ARE SPECIFIED BY BOUNDARY CONDITIONS
      IF(X.GT.0.0D0) GO TO 10
      IF(Y.EQ.Y1.OR.Y.EQ.Y2) THEN
        IF(Z.GT.Z1.AND.Z.LT.Z2) CN=0.50D0
        IF(Z.EQ.Z1.OR.Z.EQ.Z2) CN=0.25D0
      END IF
      IF(Z.EQ.Z1.OR.Z.EQ.Z2) THEN
        IF(Y.GT.Y1.AND.Y.LT.Y2) CN=0.50D0
      END IF
      IF(Y.GT.Y1.AND.Y.LT.Y2.AND.Z.GT.Z1.AND.Z.LT.Z2) CN=1.0D0
      RETURN
C
C        START NUMERICAL INTEGRATION LOOP
10    SUM=0.0D0
      DO 20 I=1,NMAX
C
C        SCALE THE GAUSS-LEGENDRE COEFFICIENTS TO ACCOUNT FOR THE
C        NON-NORMALIZED LIMITS OF INTEGRATION
C        LIMITS OF INTEGRATION ARE FROM 0 TO T**0.25
      TT=T**0.250D0
      WI=WN(I)
      ZI=TT*(ZN(I)+1.0D0)/2.0D0
```

```
      ZSQ=ZI*ZI
      Z4=ZSQ*ZSQ
C
C       TERM 1
      XVT=X-VX*Z4
      EXP1=-XVT*XVT/(4.0D0*DX*Z4)-DK*Z4
      ERFC1=(Y1-Y)/(2.0D0*ZSQ*DSQRT(DY))
      CALL EXERFC(EXP1,ERFC1,Q1)
C
C       TERM 2
      ERFC2=(Y2-Y)/(2.0D0*ZSQ*DSQRT(DY))
      CALL EXERFC(EXP1,ERFC2,Q2)
C
C       TERM 3
      EXP2=0.0D0
      ERFC1=(Z1-Z)/(2.0D0*ZSQ*DSQRT(DZ))
      CALL EXERFC(EXP2,ERFC1,Q3)
      ERFC2=(Z2-Z)/(2.0D0*ZSQ*DSQRT(DZ))
      CALL EXERFC(EXP2,ERFC2,Q4)
      TERM=(Q1-Q2)*(Q3-Q4)*WI/(ZI*ZSQ)
      SUM=SUM+TERM
20    CONTINUE
      SUM=SUM*TT/2.0D0
      CN=SUM*X/(2.0D0*DSQRT(PI*DX))
      RETURN
      END
```